

5ª Aula Prática – LCE 602 – ESTATÍSTICA EXPERIMENTAL – 2015 DELINEAMENTO INTEIRAMENTE CASUALIZADO – PRESSUPOSIÇÕES DO MODELO

Um pesquisador pretende comparar quatro variedades de pêssego quanto ao enraizamento de estacas. Para tanto, realizou um experimento de acordo com o delineamento inteiramente casualizado com cinco repetições, sendo cada parcela um vaso com vinte estacas. Passado o tempo necessário, o pesquisador anotou o número de estacas enraizadas, apresentado na Tabela 1.

Tabela1: Número de estacas enraizadas

Variedades	Repetições					Total
	1	2	3	4	5	
A	2	2	1	1	0	6
B	1	0	0	1	1	3
C	12	10	14	17	11	64
D	7	9	15	8	10	49

Realizar a análise da variância da forma mais adequada.

Aula no R

```
# Removendo todos os objetos
rm(list=ls(all=TRUE))
# Entrada dos dados para análise

y<- c( 2,  2,  1,  1,  0,
      1,  0,  0,  1,  1,
      12, 10, 14, 17, 11,
      7,  9, 15,  8, 10)
trat<- rep(c("A","B","C","D"), each=5)
dados<- data.frame(trat, y)

# Análise de variância
modelo=lm(y ~ trat, dados)
# 1. Obtenção dos resíduos e análise
# exploratória dos resíduos
res <- residuals(modelo)# resíduos simples
res_Stud <- rstandard(modelo)# resíduos
# Studentizados
round(data.frame(res,res_Stud),5)
boxplot(res_Stud)
# 2. Normalidade dos erros
# 2.1 Análise gráfica dos resíduos - Gráfico
quantil-quantil
qqnorm(res_Stud,col="blue",cex=0.7,
xlab="Quantis
      da distribuição normal",ylab="Resíduos
      Studentizados")
abline(0,1,col=2)
# 2.2 Teste de normalidade de Shapiro-Wilk
shapiro.test(res_Stud)
# 3. Homogeneidade de variâncias
# 3.1 Análise gráfica dos resíduos
library(lattice)
dotplot(res_Stud~trat,ylab="Resíduos
      Studentizados",
      xlab="Variedade")
boxplot(res_Stud~trat,ylab="Resíduos
      Studentizados",xlab="Variedade")
# 3.2 Teste de # 3.2 Teste de Breusch-Pagan
# 3.2 Teste de Breusch-Pagan

library(lmtest)
bpptest(modelo)
# 4. Verificando a necessidade de se
transformar
# os dados
# 4.1 Análise gráfica dos resíduos
plot(res_Stud~fitted(modelo),ylab="Resíduos
      Studentizados",xlab="Valores esperados
      (médias)")
abline(h=0)
# 4.2 Transformação de Box-Cox
library(MASS)
boxcox(y+0.001 ~ trat,ylab="logaritmo da
      verossimilhança")
# transformação sugerida - > y^0.5
dados$yt<- (y+0.01)^0.5
modelot<- lm(yt ~ trat, dados)
qqnorm(rstandard(modelot),col="blue",cex=0.7,
xlab="Quantis
      da distribuição normal",ylab="Resíduos
      Studentizados")
shapiro.test(rstandard(modelot))
dotplot(rstandard(modelot)~trat,ylab="Resíduos
      Studentizados",
      xlab="Variedade")
bpptest(modelot)
plot(rstandard(modelot)~fitted(modelot),ylab="Resíduos
      Studentizados",xlab="Valores esperados
      (médias)")
abline(h=0)
boxcox(modelot,ylab="logaritmo da
      verossimilhança")
anova(modelot)
library(ExpDes.pt)
dic(dados$trat, dados$yt)
#apresentação das médias
round((tapply(dados$yt, dados$trat, mean))^2-
0.01,4)
```

Aula no SAS

```
ODS HTML;
TITLE 'DIC - Pressuposições do modelo';
DATA estaca;
INPUT trat$ y @@;
DATALINES;
A 2 A 2 A 1 A 1 A 0
B 1 B 0 B 0 B 1 B 1
C 12 C 10 C 14 C 17 C 11
D 7 D 9 D 15 D 8 D 10
;
PROC SORT DATA=estaca; By trat; RUN;
TITLE 'Análise de variância e obtenção dos resíduos';
PROC GLM DATA=estaca;
CLASS trat;
MODEL y=trat/SS3;
OUTPUT OUT=residuos PREDICTED=y_est
RESIDUAL=res_ord STUDENT=res_Stud;
RUN;
PROC PRINT DATA=residuos;
RUN;
TITLE 'Verificação da Normalidade dos resíduos, Normal-plot e Teste de Shapiro-Wilk';
PROC UNIVARIATE DATA=residuos PLOT NORMAL;
VAR res_Stud; QQPLOT;
RUN;
TITLE 'Análise gráfica dos resíduos';
PROC PLOT DATA=residuos;
PLOT res_Stud*trat="*"/vpos=20;
RUN;
TITLE 'Teste de Brown-Forsythe para homogeneidade de variâncias';
PROC GLM DATA=estaca;
CLASS trat;
MODEL y = trat/SS3;
MEANS trat/HOVTEST=bf;
RUN;
TITLE 'Análise gráfica dos resíduos e Transformação de Box-Cox';
PROC PLOT DATA=residuos;
PLOT res_Stud*y_est="*"/vpos=20;
RUN;
*ODS GRAPHICS ON;
DATA estaca;
SET estaca;
yc = y+0.01;
PROC TRANSREG DATA=estaca plots(UNPACK);
MODEL BOXCOX(yc/CONVENIENT LAMBDA=-2 to 2 By 0.05) = CLASS(trat);
*ODS selectBoxCoxLogLikePlot;
RUN;
*ODS GRAPHICS OFF;
ODS HTML CLOSE;

TITLE 'Análise de variância e obtenção dos resíduos - dados transformados';
DATA estaca;
SET estaca;
yt = (y+0.01)**0.5;
PROC GLM DATA=estaca;
CLASS trat;
MODEL yt=trat/SS3;
OUTPUT OUT=residuoest PREDICTED=yt_est
RESIDUAL=res_ordt STUDENT=res_Studt;
RUN;
TITLE 'Verificação da Normalidade dos resíduos, Normal-plot e Teste de Shapiro-Wilk';
PROC UNIVARIATE DATA=residuoest PLOT NORMAL;
VAR res_Studt; QQPLOT;
RUN;
TITLE 'Análise gráfica dos resíduos';
PROC PLOT DATA=residuoest;
PLOT res_Studt*trat="*"/vpos=20;
RUN;
TITLE 'Teste de Levene para homogeneidade de variâncias';
DATA residuoest;
set residuoest;
res_abst=abs(res_ordt);
PROC GLM DATA=residuoest;
CLASS trat;
MODEL res_abst = trat/SS3;
RUN;
TITLE 'Teste de Brown-Forsythe para homogeneidade de variâncias';
PROC GLM DATA=estaca;
CLASS trat;
MODEL yt = trat/SS3;
MEANS trat/HOVTEST=bf;
RUN;
TITLE 'Análise gráfica dos resíduos e Transformação de Box-Cox';
PROC PLOT DATA=residuoest;
PLOT res_Studt*yt_est="*"/vpos=20;
RUN;
*ODS GRAPHICS ON;
PROC TRANSREG DATA=estaca plots(UNPACK);
MODEL BOXCOX(yt/CONVENIENT LAMBDA=-2 to 2 BY 0.05) = CLASS(trat);
*ODS selectBoxCoxLogLikePlot;
RUN;
*ODS GRAPHICS OFF;
PROC GLM DATA=estaca;
CLASS trat;
MODEL yt = trat/SS3;
MEANS trat/TUKEY;
RUN;
```

Exercício a ser entregue

Verificar as pressuposições da análise da variância do exercício da aula 2.